An average model for the phase-shifted converter

Christophe Basso

May 2009

Revised April 2010

Introduction

- A small-signal model has been described by Fu-Sheng Tsai in 1993
- Despite all efforts, I could not make it work (sorry about that).
- Based on the PWM switch, I modified the auto-toggling model described

in my last book to cope with phase-shifted requirements.

- The model is now fully auto-toggling between CCM and DCM.

Fu-Sheng Tsai, «Small-signal and transient analysis of a zero-voltage switched, Phase-Controlled PWM Converter Using Averaged Switch Model», IEEE Transactions, Vol. 29, n°3, May/June 1993

Cycle-by-cycle simulation to understand the signals

- ❑ The switching model uses the simplied version of a phase shift controler

parameters

fc=1k
pm=60
Gfc=-7
pfc=-81
Vout=12

G=10^(-Gfc/20)
boost=pm-(pfc)-90
pi=3.14159
K=tan((boost/2+45)*pi/180)

Fzero=fc/k
Fpole=fc*k

RLED=CTRmin*Rpullup/G
Czero=1/(2*pi*Fzero*Rupper)
CpoleX=1/(2*pi*Fpole*Rpullup)
Cpole=CpoleX-Copto

VTL431min=2.5
Vf=1
Vdd=5V
Vcesat=300m
Ibias=1m

CTRmin=0.3
Pole=4k
Copto=1/(2*pi*Pole*Rpullup)

Rpullup=4.7k
Rupper=(Vout-2.5)/250u
RLED1=Vout-Vf-VTL431min
RLED2=Vdd-Vcesat+Ibias*CTRmin*Rpullup
RLEDmax=(RLED1/RLED2)*Rpullup*CTRmin

parameters

DT=250n
CDT=DT/(0.69*1k)
Fsw=500k



Resulting signals from the switching simulation



Equations to derive the leakage inductance reset time

Voltage across the transformer primary



$$D_{off} + D = 1$$

$$D_{off} = 1 - D$$

$$D = \Delta D + D_{eff}$$

$$D_{eff} = \frac{V_{out}}{nV_{in}}$$

Output inductor equations

$$\Delta I_L = \frac{V_{out}\left(1 - D_{eff}\right)}{LF_{sw}}$$

$$I_{L,peak} = n\left(I_{out} + \frac{\Delta I_L}{2}\right) = n\left(I_{out} + \frac{V_{out}\left[1 - D_{eff}\right]}{2F_{sw}L}\right)$$

$$I_{L,valley} = n\left(I_{out} - \frac{\Delta I_L}{2}\right) = n\left(I_{out} - \frac{V_{out}\left[1 - D_{eff}\right]}{2F_{sw}L}\right)$$

F_{sw} is the output ripple period which is twice the primary leakage period

Equations to derive the leakage inductance reset time

$$I_1 = I_{L,peak} - D_{off} T_{sw} S_f = I_{L,peak} - (1-D) n T_{sw} \frac{V_{out}}{L}$$

$$I_1 = n\left(I_{out} + \frac{\Delta I_L}{2}\right) - (1-D) n T_{sw} \frac{V_{out}}{L}$$

$$I_1 = n\left(I_{out} + \frac{V_{out}\left(1-D_{eff}\right)}{2LF_{sw}} - (1-D) T_{sw} \frac{V_{out}}{L}\right)$$

The current excursion across the leakage inductor is $\Delta I = I_{L,valley} + I_1$

The voltage excursion across the leakage inductor is V_{in}

The current slope across the leakage inductor is V_{in}/l_{leak}

The leakage inductor reset time is then: $t = \Delta I \, l_{leak}/V_{in}$

$$\Delta I = n\left(I_{out} - \frac{V_{out}\left[1-D_{eff}\right]}{2F_{sw}L}\right) + n\left(I_{out} + \frac{V_{out}\left(1-D_{eff}\right)}{2LF_{sw}} - (1-D) T_{sw} \frac{V_{out}}{L}\right)$$

Equations to derive the leakage inductance reset time

$$\Delta D = \frac{\Delta I}{T_{sw}\dfrac{V_{in}}{l_{leak}}}$$

$$\Delta D = \frac{n\left(I_{out} - \dfrac{V_{out}\left[1-D_{eff}\right]}{2F_{sw}L_{out}}\right) + n\left(I_{out} + \dfrac{V_{out}\left(1-D_{eff}\right)}{2L_{out}F_{sw}} - \left(1-D\right)T_{sw}\dfrac{V_{out}}{L}\right)}{\dfrac{V_{in}}{l_{leak}}T_{sw}}$$

$$\Delta D = \frac{n\left[2I_{out} - T_{sw}\dfrac{V_{out}}{L}\left(1-D\right)\right]}{\dfrac{V_{in}}{l_{leak}}T_{sw}}$$

$$\Delta D = \frac{n}{\dfrac{V_{in}}{L_{lk}}\dfrac{T_s}{2}}\left(2I_L - \frac{V_{out}}{L}(1-D)\frac{T_s}{2}\right)$$

The two formulas exactly match, T_s is the controller switching period whereas T_{sw} is the output ripple period: $T_s = 2\,T_{sw}$

V. Vlatkcovic and al., «Small-Signal Analysis of the Phase Shifted PWM Converter», IEEE transactions, Vol. 7, n°1, January 1992

Equations to derive the leakage inductance reset time

- Numerical application:

$$n = \frac{1}{6},\ I_{out} = 24\ A,\ F_{sw} = 500\ kHz,\ l_{leak} = 15.7\ \mu H$$

$$L = 3.43\ \mu H,\ V_{in} = 240\ V,\ V_{out} = 12\ V$$

$$\Delta D = \frac{n\left[2I_{out} - T_{sw}\frac{V_{out}}{L}(1-D)\right]}{\frac{V_{in}}{l_{leak}}T_{sw}} = 0.245$$

$$\Delta D T_{sw} = 490\ ns$$ → Simulation gives 491 ns

$$D = \Delta D + D_{eff} = \Delta D + \frac{V_{out}}{nV_{in}} = 0.245 + \frac{12}{166m \times 240} = 0.54$$

$$I_{L,peak} = 4.408\ A \qquad I_{L,valley} = 3.591\ A \qquad I_1 = 3.9\ A$$

In the PWM switch model, the ripple is neglected

- In the previous expression, the leakage inductor current swinged between the peak value and I_2.
- On average, we consider the inductor current ripple free
- Therefore, the leakage current swings between $-nI_{out}$ and $+nI_{out}$
- In the buck average model, $I_{out} = I_c$.
- The average equation for ΔD is therefore:

$$\Delta D_{avg} = \frac{2 l_{leak} I_{out} F_{sw}}{n V_{in}} \longrightarrow \text{Gives } 0.262$$

- In the PWM switch, I_c is the output current terminal and V_{in} is applied across V_{ap}:

$$\Delta D_{avg} = \frac{2 l_{leak} I_c F_{sw}}{n V_{ap}}$$

Conduction Mode Transition Point Calculations

- Discontinuous Conduction Mode, classical buck:

The critical point is the same as for a classical buck except that D_{eff} must be used:

$$R_{crit} = 2LF_{sw}\frac{nV_{in}}{nV_{in} - V_{out}} = 2\times 3.47u\times 1Meg\,\frac{0.166\times 240}{0.166\times 240 - 12} = 9.9\ \Omega$$

This is the output ripple frequency or twice the oscillator value.

- Model transition point:



$$D_{off} = 1 - D_{eff}$$

$$D_{off} = 1 - (D - \Delta D)$$

$$D_{off} = 1 - D + \Delta D$$

Conduction Mode Transition Point Calculations

❑ The mode transition can also be derived by cancelling the equation describing ΔD:

$$\Delta D + \frac{V_{out}}{nV_{in}}$$

$$\Delta D = \frac{n\left[2I_{out} - T_{sw}\frac{V_{out}}{L}(1-D)\right]}{\frac{V_{in}}{l_{leak}}T_{sw}} = 0$$

$$\Delta D = \frac{l_{leak}\left(T_{sw}V_{out}{}^{2} - nT_{sw}V_{out}V_{in} + 2nI_{out}LV_{in}\right)}{T_{sw}V_{in}\left(LV_{in} - nl_{leak}V_{out}\right)} = 0$$

$$l_{leak}\left(T_{sw}V_{out}{}^{2} - nT_{sw}V_{out}V_{in} + 2nI_{out}LV_{in}\right) = 0$$

$$\longrightarrow \quad I_{out,crit} = \frac{T_{sw}V_{out}\left(nV_{in} - V_{out}\right)}{2nLV_{in}}$$

$$R_{crit} = \frac{V_{out}}{I_{out,crit}} = \frac{2nLV_{in}}{T_{sw}\left(nV_{in} - V_{out}\right)} == \frac{2\times 0.166\times 3.47u\times 240}{1u\times(0.166\times 240 - 12)} = 9.9\ \Omega$$

- The phase-shifted converter has been loaded by a 9.9-Ω resistor
- The switching ripple on the output inductor is 1 MHz



- A formula including the effect of the leakage inductance has been derived by Monsieur Schutten from General Electric:

$$R_{crit} = 2F_{sw} \frac{\left(l_{leak} n^2 + L\right)}{1 - \frac{V_{out}}{nV_{in}}}$$

- It is similar to those derived before except that the total leakage (reflected to the secondary side) is accounted in series with the output inductor.

- For large turns ratios, the contribution of the reflected leakage term is weak. However, for smaller ratios, it can make a difference between the results delivered by the first set of equations and Mr Schutten’s equation.

Conduction Mode Transition Point Calculations



$D + D_{off} = 1$ CCM $1 : \boxed{D - \Delta D} \rightarrow D_{eff}$

$\Delta D = 0$ DCM $1 : \dfrac{D}{D + D_{off}}$

To match
the book notation: $1 : \dfrac{d_1 - \Delta D}{d_1 + d_2}$

In the DCM PWM switch model described page 166 of my book:

$d_2 = \dfrac{2LF_{sw}}{D}\dfrac{I_c}{V_{ac}} - d_1$ d_2 is the off duty-cycle in DCM, d_1 is the on duty-cycle

When d_2 hits $1 - d_1 + \Delta D$ ⟶ The model leaves DCM and enters CCM

By clamping d_2 between 0 and $1 - d_1 + \Delta D$ the models auto-toggles.

Average model implementation

```
.SUBCKT PWMswitchPS a c p d dL params: L=10u Ll=1u Fs=100k
*
* This subckt is DCM-CCM phase shift average model
*
* 1 - active ; 2 - passive ; 3 - common ; 4 - duty-cycle ; 5 - dL
*
.subckt limit d dc params: clampH=0.99 clampL=16m
Gd 0 dcx d 0 100u
Rdc dcx 0 10k
V1 clpn 0 {clampL}
V2 clpp 0 {clampH}
D1 clpn dcx dclamp
D2 dcx clpp dclamp
Bdc dc 0 V=V(dcx)
.model dclamp d n=0.01 rs=100m
.ENDS
*
.subckt limit2 d2nc d d2c dL
Gd 0 d2cx d2nc 0 100u
Rdc d2cx 0 10k
V1 clpn 0 7m
BV2 clpp 0 V=1-V(d)+V(dL)-6.687m
D1 clpn d2cx dclamp
D2 d2cx clpp dclamp
B2c d2c 0 V=V(d2cx)
.model dclamp d n=0.01 rs=100m
.ENDS
*
Xd d dc limit params: clampH=0.99 clampL=16m
BVcp 6 p V=((V(dc)-V(dL))/(V(dc)-V(dL)+V(d2)+1u))*V(a,p)
BIap a p I=((V(dc)-V(dL))/(V(dc)-V(dL)+V(d2)+1u))*I(VM)
Bd2 d2X 0 V=(2*I(VM)*{L}-v(a,c)*V(dc)^2*{1/Fs}) / ( v(a,c)*V(dc)*{1/Fs}+1u )
Xd2 d2X dc d2 dL limit2
BdL dL 0 V=2*{Ll}*I(VM)*{Fs}/(V(a,p)+1u) < 0 ? 0 : 2*{Ll}*I(VM)*{Fs}/(V(a,p)+1u)
VM 6 c
*
.ENDS
```

Leakage inductance

Accounts for ΔD

$$d_2 = \frac{2I_c L - V_{ac} d_1^{\,2} T_{sw}}{V_{ac} d_1 T_{sw}}$$

Clamp of d_2

$$\Delta D_{avg} = \frac{2 l_{leak} I_c F_{sw}}{n V_{ap}}$$

Average model implementation, IsSpice

Calculations give 0.245



Ac simulations versus prototype values



Control to output

M. Schutten and al., "Improved Small-Signal Analysis for the Phase-Shifted PWM Power Converter", IEEE Transactions, Vol. 18, n°2, 2003

Ac simulations versus prototype values



Audio susceptibility

M. Schutten and al., “Improved Small-Signal Analysis for the Phase-Shifted PWM Power Converter”, IEEE Transactions, Vol. 18, n°2, 2003

Mode transition point

- The load resistor is increased until the ac response changes



Transient simulations versus cycle-by-cycle results



Conclusion

- A simple and efficient average model has been derived
- It automatically toggles between CCM and DCM
- It predicts the mode transition point
- It matches, in CCM, ac prototype measurements
- It can easily be ported to different simulator languages